6.5 Exponentially Distributed Trajectory Lengths

In[582]:=
```
ClearGlobal[] := (ClearAll["Global`*"]; Clear[Derivative];);
RemoveGlobal[] := (ClearGlobal[]; Remove["Global`*"];);
Clear[r, fn, B, g, a, b, c, d, a2, b2, c2, d2, τ, β, y, j, k, ν, μ, p];
Clear[fn00, fn11, fn20, fn02, g00, g11, g02, g20];
Clear[gt00, gt11, gt02, gt20, temp, gt];
Clear[expExpr, binomExpr, factored, unfactored, unfactoredTrig]
Clear[vars, rule, check2]
Clear[transform, transform0]
Clear[intermediate, intermediate1,
 intermediate2, intermediate3, intermediate4]
```

Laplace Transform (Integral)

This is the bare function that is presented in the Laplace Transform of Equation (41) in the paper with the exponentially distributed trajectory lengths included,

In[591]:=
```
(*p := Piecewise[{{1/τ E^(- y/τ),y≥0}},0]*)
(*This\ explicit\ form\ just\ drains\ computational\ time!*)
p := 1 / τ E^(-y/τ)
fn[j_, k_, y_, β_, τ_, ω_] := p Cos[ω y]^j Sin[ω y]^k
```

Attempted Solution (Integrand)

This is an attempt to recreate Equation (43). Expanding into exponentials,

In[593]:=
```
expExpr = TrigToExp[fn[j, k, y, β, τ, ω]]
```

Out[593]=
$$\frac{2^{-j-k}\, e^{-\frac{y}{\tau}} \left(i\left(e^{-i y \omega} - e^{i y \omega}\right)\right)^{k} \left(e^{-i y \omega} + e^{i y \omega}\right)^{j}}{\tau}$$

and then expanding with binomial coefficients,

In[594]:=
$$\texttt{binomExpr} = \frac{1}{\tau} (-1)^{\nu}\, 2^{-j-k}\, i^{k}\, \texttt{Sum}\left[\texttt{Sum}\left[\texttt{Binomial}[j, \mu]\, \texttt{Binomial}[k, \nu]\, e^{-i y \omega (j+k-2\mu-2\nu)-\frac{y}{\tau}}, \{\nu, 0, k\}\right], \{\mu, 0, j\}\right]$$

Out[594]=
$$\frac{(-1)^{\nu}\, i^{k}\, 2^{-j-k}\, e^{-\frac{y}{\tau}-i (j+k) y \omega} \left(1 + e^{2 i y \omega}\right)^{j+k}}{\tau}$$

- Sanity Check 1: Ensure factorisation of exponential terms are correct

In[595]:=
$$\texttt{factored} := e^{-i y \omega (j+k-2\mu-2\nu)-\frac{y}{\tau}};$$
$$\texttt{unfactoredTrig} := e^{-i y \omega (j-\mu)}\, e^{+i y \omega \mu}\, e^{-i y \omega (k-\nu)}\, e^{+i y \omega \nu};$$
$$\texttt{unfactored} := e^{-\frac{y}{\tau}}\, \texttt{unfactoredTrig};$$

In[598]:=
```
FullSimplify[factored == unfactored]
```
Out[598]=
```
True
```

- Sanity Check 2 : Check origin form can be recovered to ensure binomial expansion was correct

In[599]:=
```
check2 =
  Sum[Sum[Binomial[j, μ] Binomial[k, ν] unfactoredTrig (-1)^ν (1/2)^j (i/2)^k, {ν, 0, k}],
   {μ, 0, j}];
```

In[600]:=
$$\text{intermediate} = \left(\frac{i}{2} e^{-i y \omega} \left(1 - e^{2 i y \omega}\right)\right)^k \left(\frac{1}{2} e^{-i y \omega} \left(1 + e^{2 i y \omega}\right)\right)^j;$$

In[601]:=
$$\text{FullSimplify}\left[\frac{1}{\tau} e^{-\frac{y}{\tau}} \text{ExpToTrig[intermediate]} == \text{expExpr}\right]$$

Out[601]=
```
True
```

The relevant parts can be extracted and simplified for the Laplace Transform,

In[602]:=
$$\text{transform0} = \text{LaplaceTransform[factored, y, } \beta] \frac{1}{\tau} (-1)^\nu 2^{-j-k} i^k \text{ // FullSimplify}$$

Out[602]=
$$\frac{(-1)^\nu i^k 2^{-j-k}}{1 + \beta \tau + i (j + k - 2 (\mu + \nu)) \tau \omega}$$

and can be brought into the familar form of Equation (43),

In[603]:=
$$\text{transform}[\nu\_, \mu\_, j\_, k\_] := \frac{(-1)^{\nu + \frac{1}{2} k} \left(\frac{1}{2}\right)^{j+k}}{1 + \beta \tau + i \tau \omega (j + k - 2 \mu - 2 \nu)};$$

In[604]:=
```
FullSimplify[transform[ν, μ, j, k] == transform0]
```
Out[604]=
```
True
```

and this gives the final form of the Integrand,

In[605]:=
```
gt[j_, k_] := Sum[
   Sum[Binomial[j, μ] Binomial[k, ν] transform[ν, μ, j, k], {ν, 0, k}], {μ, 0, j}]
```

Paper Solution (Integrand)

Equation (43) is the integrand of the integral in Equation (41) with exponentially distributed trajectory lengths,

In[606]:=
```
B[k_] := If[IntegerQ[k / 2], β τ + 1, (j + k - 2 μ - 2 ν) ω τ];
g [j_, k_] :=
```
$$\text{Sum}\left[\text{Sum}\left[\text{Binomial}[j, \mu] \text{ Binomial}[k, \nu] \frac{\left(\frac{1}{2}\right)^{j+k} (-1)^{\nu + k/2} B[k]}{(\beta \tau + 1)^2 + (\omega \tau)^2 (j + k - 2 \mu - 2 \nu)^2}, \{\nu, 0, k\}\right], \{\mu, 0, j\}\right]$$

Comparing Integral and Integrand (Paper form)

There are only four specific cases of `j,k` that are actually used in the paper. Hence it makes sense to just assume that the domain of the solution is restricted to these cases for simplicity.

Integral Forms

In[608]:=
```
fn00 := LaplaceTransform[fn[0, 0, y, β, τ, ω], y, β] // FullSimplify
fn11 := LaplaceTransform[fn[1, 1, y, β, τ, ω], y, β] // FullSimplify
fn20 := LaplaceTransform[fn[2, 0, y, β, τ, ω], y, β] // FullSimplify
fn02 := LaplaceTransform[fn[0, 2, y, β, τ, ω], y, β] // FullSimplify
```

Integrand Forms

In[612]:=
```
g00 := FullSimplify[g[0, 0]]
g11 := FullSimplify[g[1, 1]]
g20 := FullSimplify[g[2, 0]]
g02 := FullSimplify[g[0, 2]]
```

Agreement of Paper Integral vs. Integrand

The following three are in agreement,

In[616]:=
```
FullSimplify[fn00 == g00]
FullSimplify[fn20 == g20]
FullSimplify[fn02 == g02]
```

Out[616]= True

Out[617]= True

Out[618]= True

Disagreement of Paper Integral vs. Integrand

The final case of `j=k=1` gives a difference in a factor of *i*,

In[619]:=
```
fn11 // FullSimplify
g11 // FullSimplify
```

Out[619]= $\frac{\tau\,\omega}{(1+\beta\,\tau)^2+4\,\tau^2\,\omega^2}$

Out[620]= $\frac{i\,\tau\,\omega}{(1+\beta\,\tau)^2+4\,\tau^2\,\omega^2}$

Resolution

The difference can be resolved by an addition of - *i* in the value B for $\frac{1}{2}$k ∈ {0,1,...,n},

In[621]:=
```
B[k_] := If[IntegerQ[k / 2], β τ + 1, -i (j + k - 2 μ - 2 ν) ω τ];
```

In[622]:=
```
FullSimplify[fn00 == g00]
FullSimplify[fn11 == g11]
FullSimplify[fn20 == g20]
FullSimplify[fn02 == g02]
```

Out[622]= True

Out[623]= True

Out[624]= True

Out[625]= True

Comparing Integral and Integrand (Attempted form)

There are only four specific cases of **j,k** that are actually used in the paper. Hence it makes sense to just assume that the domain of the solution is restricted to these cases for simplicity.

Integral Forms

Defined already above.

Integrand Forms

In[626]:=
```
gt00 := FullSimplify[gt[0, 0]]
gt11 := FullSimplify[gt[1, 1]]
gt20 := FullSimplify[gt[2, 0]]
gt02 := FullSimplify[gt[0, 2]]
```

Agreement

All agree

In[630]:=
```
FullSimplify[fn00 == gt00]
FullSimplify[fn11 == gt11]
FullSimplify[fn20 == gt20]
FullSimplify[fn02 == gt02]
```

Out[630]= True

Out[631]= True

Out[632]= True

Out[633]= True

Appendix

Need to take baby steps to let Mathematica catch up with this simplification else it fails. The first expression is validated against **check2** and the final expression is validated against **intermediate** which was then validated to be equal to **check2**

In[634]:= intermediate1 = $i^{k}\, e^{-i\, y\, k\, \omega} \left(1 - e^{2\, i\, y\, \omega}\right)^{k} \left(\frac{1}{2}\right)^{j+k} e^{-i\, j\, y\, \omega} \left(1 + e^{2\, i\, y\, \omega}\right)^{j}$;

In[635]:=
```
FullSimplify[intermediate1 == check2,
 Element[k, Integers] && Element[j, Integers] && j ≥ 0 && k ≥ 0]
```

Out[635]= True

In[636]:= intermediate2 = $i^{k}\, e^{-i\, y\, \omega\, k} \left(1 - e^{2\, i\, y\, \omega}\right)^{k} \left(\frac{1}{2}\right)^{j+k} \left(e^{-i\, y\, \omega} \left(1 + e^{2\, i\, y\, \omega}\right)\right)^{j}$;

In[637]:=
```
FullSimplify[intermediate2 == intermediate1,
 Element[k, Integers] && Element[j, Integers] && j ≥ 0 && k ≥ 0]
```

Out[637]= True

In[638]:=
```
intermediate3 = ⅈ^k (ⅇ^(-ⅈ γ ω) (1 - ⅇ^(2 ⅈ γ ω)))^k (1/2)^(j+k) (ⅇ^(-ⅈ γ ω) (1 + ⅇ^(2 ⅈ γ ω)))^j;
```

$$\text{intermediate3} = i^k \left(e^{-i\gamma\omega}\left(1 - e^{2 i \gamma \omega}\right)\right)^k \left(\frac{1}{2}\right)^{j+k} \left(e^{-i\gamma\omega}\left(1 + e^{2 i \gamma \omega}\right)\right)^j;$$

In[639]:=
```
FullSimplify[intermediate3 == intermediate2,
 Element[k, Integers] && Element[j, Integers] && j ≥ 0 && k ≥ 0]
```

Out[639]= True

In[640]:=
```
FullSimplify[intermediate == intermediate3,
 Element[k, Integers] && Element[j, Integers] && j ≥ 0 && k ≥ 0]
```

Out[640]= True

In[641]:=